

CHAPITRE 1

Visualisation, Fractales

1. Généralités

1.1 Définition

Une figure fractale est une courbe ou surface de forme irrégulière ou morcelée qui se crée en suivant des règles déterministes ou stochastiques impliquant une homothétie interne. Le terme fractale est un néologisme créé par Benoît Mandelbrot en 1974 à partir de la racine latine « fractus », qui signifie brisé, irrégulier. Dans la théorie de la rugosité développée par Mandelbrot, une fractale désigne des objets dont la structure est invariante par changement d'échelle.

Des formes fractales approximatives sont facilement observables dans la nature. Ces objets ont une structure autosimilaire sur une échelle étendue, mais finie : les nuages, les flocons de neige, les montagnes, les réseaux de rivières, le chou-fleur ou le brocoli, et les vaisseaux sanguins, etc.

Les domaines d'application des fractales sont très nombreux, on peut citer en particulier :

- En biologie, répartition des structures des plantes, bactéries, feuilles, branches d'arbres...
- En médecine, structure des poumons, intestins, battements du cœur
- En électronique, antennes larges bandes des téléphones portables
- En météorologie, nuages, vortex, banquise, vagues scélérates, turbulences, structure de la foudre
- En astronomie avec la description des structures de l'univers, cratères sur la Lune, répartition des exoplanètes et des galaxies...

1.2 Quelques propriétés

Propriété 1.2.1 *Quand la fractale est formée de répliques d'elle-même en plus petit, sa dimension fractale peut se calculer comme suit :*

$$D = \frac{\ln(n)}{\ln(h)}$$

où la fractale de départ est formée de n exemplaires dont la taille a été réduite d'un facteur h (pour homothétie).

Quelques exemples :

- Un côté du flocon de Koch est formé de $n = 4$ exemplaires de lui-même réduit d'un facteur $h = 3$. Sa dimension fractale vaut :

$$D = \frac{\ln(4)}{\ln(3)}$$

- Le triangle de Sierpinski est formé de $n = 3$ exemplaires de lui-même réduit d'un facteur $h = 2$. Sa dimension fractale vaut :

$$D = \frac{\ln(3)}{\ln(2)}$$

Propriété 1.2.2 Si on obtient une fractale à partir d'une figure vivant dans un espace ambiant de dimension N , on a : $N - 1 < D < N$, avec D étant la dimension de la figure obtenue.

Définition 1.2.3 Un IFS est une famille S de N fonctions contractantes $T_i : M \rightarrow M$ dans un espace métrique complet M . On définit à partir des T_i une nouvelle fonction T , elle aussi contractante sur l'ensemble des parties compactes de M muni de la distance de Hausdorff, par l'expression :

$$T(A) = \bigcup_{i=1}^N T_i(A).$$

- Les transformations affines du plan, sont de la forme $\begin{pmatrix} x \\ y \end{pmatrix} \rightarrow \begin{pmatrix} a & b \\ c & d \end{pmatrix} \begin{pmatrix} x \\ y \end{pmatrix} + \begin{pmatrix} e \\ f \end{pmatrix}$

La matrice $\begin{pmatrix} e \\ f \end{pmatrix}$ est une translation. La matrice $\begin{pmatrix} a & b \\ c & d \end{pmatrix}$ est une application linéaire.

- Il existe plusieurs types d'applications linéaires :
 - Lorsque $\begin{pmatrix} a & b \\ c & d \end{pmatrix} = \begin{pmatrix} r \cos \theta & -r \sin \theta \\ r \sin \theta & r \cos \theta \end{pmatrix}$ il s'agit d'une rotation.
 - Lorsque $\begin{pmatrix} a & b \\ c & d \end{pmatrix} = \begin{pmatrix} \lambda 1 & 0 \\ 0 & \lambda 2 \end{pmatrix}$ il s'agit d'une homothétie.

2. Fractales, IFS et « Chaos game »

Dans cette partie, pour les objets fractals que je vais construire, il existe à chaque fois deux manières pratiques de procéder. Elles ne donneront pas les mêmes résultats visuels mais sont toutes les deux intéressantes à expérimenter.

2.1. Méthode déterministe

Dans cette méthode, on va faire appel à la récursivité, on la qualifiera de déterministe par rapport à l'autre.

2.1.1. Triangle de Sierpinski



Algorithme : Un algorithme pour obtenir des approximations arbitrairement proches du triangle de Sierpinski peut s'écrire de la manière récurrente suivante :

1. Commencer à partir d'un triangle quelconque du plan. Le triangle canonique de Sierpinski se construit à partir d'un triangle équilatéral ayant une base parallèle à l'axe des abscisses.
2. Tracer les trois segments qui joignent deux à deux les milieux des côtés du triangle, ce qui délimite 4 nouveaux triangles.
3. Enlever le petit triangle central. Il y a maintenant trois petits triangles qui se touchent deux à deux par un sommet, dont les longueurs des côtés sont la moitié de celles du triangle de départ (obtenue par une homothétie de rapport 1/2), et dont l'aire est divisée par 4.
4. Recommencer à la deuxième étape avec chacun des petits triangles obtenus.


Démarche : Pour cette fractale, j'ai écrit une fonction récursive acceptant en entrée les coordonnées de trois points dans les vecteurs colonnes A, B, C et renvoyant dans triangles une matrice à deux lignes contenant sur chaque colonne les coordonnées (x, y) des points successifs de tous les triangles pleins de domaine intérieur du triangle (A,B,C).

Par exemple, en prenant le triangle (A, B, C) où $A(0,0)$, $B(1,0)$, $C(0.5, \sqrt{3}/2)$, On réalise le programme suivant :

```
function triangles=sierpinski(A, B, C, n)
    if n == 0 then
        triangles = [A,B,C];
    else
        D = (A + B) / 2;
        E = (A + C) / 2;
        F = (B + C) / 2;
        Triangles = [sierpinski(C,E,F,n-1), sierpinski(A,E,D,n-1), sierpinski(F,D,B,n-1)];
    end
endfunction

A = [0;0];
B = [1;0];
C = [0.5;sqrt(3)/2];
tri = sierpinski(A,B,C,8)

clf
x = matrix(tri(1,:),3,-1);
y = matrix(tri(2,:),3,-1);
plot3d(x,y,0*x);
set(gca(),"view","2d")
```



Le triangle de Sierpinsky obtenue à l'aide de méthode déterministe

2.1.2. Flocon de Von Kock



On considère le segment $K_0 = AE$ et on définit K_1 par la courbe union des segments AB, BC, CD, DE , où les nouveaux points sont définis par leur vecteur de coordonnées obtenus par

$$B = A + \vec{u}, \quad C = B + R\vec{u}, \quad D = E - \vec{u}$$

Avec $\vec{u} = \frac{1}{3}\overrightarrow{AB}$ et $R = \begin{pmatrix} \cos \frac{\pi}{3} & -\sin \frac{\pi}{3} \\ \sin \frac{\pi}{3} & \cos \frac{\pi}{3} \end{pmatrix}$ la matrice de rotation d'angle $\frac{\pi}{3}$.

On note G la transformation telle que $K_1 = G(K_0)$. Ensuite, pour tout n , K_{n+1} est la courbe obtenue en appliquant G à des segments de K_n .

Un code Scilab associé peut être le suivant :


```

function points=G(A, E, n)
    if (n == 0) then
        points = [A,E];
    else
        U = 1/3 * (E - A);
        B = A + U;
        R = [cos(%pi/3) -sin(%pi/3); sin(%pi/3) cos(%pi/3)];
        C = B + R * U;
        D = E - U;
        points = [G(A,B,n-1) G(B,C,n-1) G(C,D,n-1) G(D,E,n-1)];
    end;
endfunction

A = [0;0];
E = [1;0];
result = G(A,E,5)
clf
plot(result(1,:),result(2,:))
set(gca(),"view","2d")

```

Code source de flocon de Von Kock



Courbe de Koch, après 7 itérations

2.2. Méthode utilisant un IFS et le « chaos game »

IFS est l'acronyme anglo-saxon pour « Système de Fonctions itérées ». Un IFS aléatoire est un IFS où chaque transformation affine w_i est pondérée par une probabilité $0 < p_i < 1$, telle que $\sum_{i=1}^N p_i = 1$. On se donne un point $(x_0; y_0)$, on choisit au hasard et avec une probabilité p_k la transformation w_k et on construit $(x_1; y_1) = w_k(x_0; y_0)$ et on recommence. Les itérations successives conduisent au même attracteur K obtenu avec l'IFS déterministe.

2.2.1. Le « chaos game »

Le Chaos game ou jeu du chaos a été introduit en 1993 par Michael Barnsley. À l'origine, il s'agissait d'une méthode simple et rapide de création de fractales utilisant un polygone et un point initial choisi au hasard dans ce polygone. La fractale est réalisée, en créant, par itérations successives une séquence de points, partant d'un point initial choisi aléatoirement, pour lesquels chaque point de la séquence est positionné à une fraction donnée de la distance qui sépare le point précédent d'un des sommets du polygone. Ce sommet est choisi aléatoirement à chaque itération. En répétant ce processus un nombre de fois important, et en ignorant les premiers points de la suite, un motif fractal apparaît dans la plupart des cas.

On se donne n transformations affines contractantes définies par :

$$T_i(\vec{x}) = M_i \vec{x} + b_i, \quad i = 1, \dots, n$$

telles que l'ensemble fractal étudié \mathcal{E} vérifie la propriété :

$$\mathcal{E} = \bigcup_{i=1}^n T_i(\mathcal{E})$$

Puis on considère le processus aléatoire défini par un sous ensemble compact X_0 (par exemple un point) et on définit pour n donné X_{n+1} à partir de X_n tel que

$$\text{Prob}(X_{n+1} = T_i(X_n)) = p_i$$

2.2.2. Le Triangle de Sierpinski

L'IFS du Triangle de Sierpinski est :

$$M_1 = \frac{1}{2} \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix} \quad b_1 = \vec{0}$$

$$M_2 = \frac{1}{2} \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix} \quad b_2 = \begin{pmatrix} 1/2 \\ 0 \end{pmatrix}^T$$

$$M_3 = \frac{1}{2} \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix} \quad b_3 = \begin{pmatrix} 1/4 \\ \sqrt{3}/4 \end{pmatrix}^T$$

$$\text{Où } T_n(\vec{x}) = M_n \vec{x} + b_n, \quad n = 1, \dots, N$$

Voici un code Scilab calculant les termes successifs de la suite (T_n) pour $n = 1..N$ en prenant $T_0 = (0,0)$:

```

N=10000;
x = zeros(2,N);
p = [1 1 1]/3;
P = cumsum(p);

M1 = 0.5 * eye(2,2);
M2 = 0.5 * eye(2,2);
M3 = 0.5 * eye(2,2);


b1 = zeros(2,1);
b2 = [0.5;0];
b3 = [1;sqrt(3)]/4;

for i=1:N-1
    r = rand();
    if r < P(1)
        x(:,i+1) = M1 * x(:,i) + b1;
    elseif r < P(2)
        x(:,i+1) = M2 * x(:,i) + b2;
    else
        x(:,i+1) = M3 * x(:,i) + b3;
    end
end

clf
plot(x(1,:),x(2,:),',','markersize',1)

```

Code source de triangle Sierpinski en utilisant IFS



Le Triangle de Sierpinski avec $N = 10\,000$

2.2.3. La courbe de Von Kock

L'IFS de la courbe Von Kock est :

$$M_1 = \begin{pmatrix} 1/3 & 0 \\ 0 & 1/3 \end{pmatrix}$$

$$b_1 = \vec{0}$$

$$M_2 = \begin{pmatrix} 1/3 & 0 \\ 0 & 1/3 \end{pmatrix} \begin{pmatrix} \cos \frac{\pi}{3} & -\sin \frac{\pi}{3} \\ \sin \frac{\pi}{3} & \cos \frac{\pi}{3} \end{pmatrix}$$

$$b_2 = \begin{pmatrix} 1/3 \\ 0 \end{pmatrix}^T$$

$$M_3 = \begin{pmatrix} 1/3 & 0 \\ 0 & 1/3 \end{pmatrix} \begin{pmatrix} \cos \frac{-\pi}{3} & -\sin \frac{-\pi}{3} \\ \sin \frac{-\pi}{3} & \cos \frac{-\pi}{3} \end{pmatrix}$$

$$b_3 = \begin{pmatrix} 1/2 \\ \sqrt{3}/6 \end{pmatrix}^T$$

$$M_4 = \begin{pmatrix} 1/3 & 0 \\ 0 & 1/3 \end{pmatrix}$$

$$b_4 = \begin{pmatrix} 2/3 \\ 0 \end{pmatrix}^T$$


Où $T_n(\vec{x}) = M_n \vec{x} + b_n$, $n = 1, \dots, N$

```
N=5000;
x = zeros(2,N);
p = [1 1 1 1]/4;
P = cumsum(p);

M1 = 1/3 * eye(2,2);
M2 = 1/3 * eye(2,2) * [cos(%pi/3) -sin(%pi/3); sin(%pi/3) cos(%pi/3)];
M3 = 1/3 * eye(2,2) * [cos(-%pi/3) -sin(-%pi/3); sin(-%pi/3) cos(-%pi/3)];
M4 = 1/3 * eye(2,2);

b1 = zeros(2,1); b2 = [1/3;0]; b3 = [1/2;sqrt(3)/6]; b4 = [2/3;0];

for i=1 : N-1
    r = rand();
    if r < P(1)
        x(:,i+1) = M1 * x(:,i) + b1;
    elseif r < P(2)
        x(:,i+1) = M2 * x(:,i) + b2;
    elseif r < P(3)
        x(:,i+1) = M3 * x(:,i) + b3;
    else
        x(:,i+1) = M4 * x(:,i) + b4;
    end
end
```

La courbe de Von Kock avec $N = 5000$

2.2.4. La fougère de Barnsley

Barnsley propose, pour dessiner une fougère, l'IFS suivant :

$$M_1 = \begin{pmatrix} 0.00 & 0.00 \\ 0.00 & 0.16 \end{pmatrix}$$

$$b_1 = \begin{pmatrix} 0.00 \\ 0.00 \end{pmatrix}^T$$

$$M_2 = \begin{pmatrix} 0.85 & 0.04 \\ -0.04 & 0.85 \end{pmatrix}$$

$$b_2 = \begin{pmatrix} 0.00 \\ 1.6 \end{pmatrix}^T$$

$$M_3 = \begin{pmatrix} 0.20 & -0.26 \\ 0.23 & 0.22 \end{pmatrix}$$


$$b_3 = \begin{pmatrix} 0.00 \\ 1.6 \end{pmatrix}^T$$

$$M_4 = \begin{pmatrix} -0.15 & 0.28 \\ 0.26 & 0.24 \end{pmatrix}$$

$$b_4 = \begin{pmatrix} 0.00 \\ 0.44 \end{pmatrix}^T$$

Où $T_n(\vec{x}) = M_n \vec{x} + b_n$, $n = 1, \dots, N$, avec les probabilités suivant :

$$P_1 = 0.01 ; P_2 = 0.85 ; P_3 = 0.07 ; P_4 = 0.07$$



La fougère de Barnsley via un programme récursif à l'itération $n = 10\,000$

3. Ensembles de Mandelbrot et de Julia

La suite (z_n) de \mathbb{C} définie par : $z_{n+1} = z_n^2 + c$, Où z_0 et c sont donnés dans \mathbb{C} . On définit les deux ensembles suivants, tout d'abord l'ensemble de Mandelbrot, pour lequel on prend z_0 et

$$M = \{c \in \mathbb{C}, \exists M > 0, \forall n \geq 0, |z_n| < M\},$$

C'est-à-dire que M est l'ensemble des nombres complexes c pour lesquels la suite (z_n) est bornée. De manière duale, on définit l'ensemble de Julia associé à un nombre complexe c par

$$J(c) = \{z_0 \in \mathbb{C}, \exists M > 0, \forall n \geq 0, |z_n| < M\},$$

C'est-à-dire que pour c donné, $J(c)$ est l'ensemble des valeurs initiales z_0 pour lesquels la suite (z_n) est bornée.

3.1. Ensemble de Mandelbrot

3.1.1. Un résultat important

Dans cette partie, on va clarifier les choses concernant le majorant M et en déduire des informations utiles pour déterminer ces ensembles :

1. S'il existe p tel que $|z_p| > 2$ et $|z_p| > |c|$ alors $\lim_{n \rightarrow \infty} |z_n| = \infty$
2. En déduire que

- (a) Si $|c| > 2$, alors $c \notin M$
 (b) Si $|c| \leq 2$, alors s'il existe p tel que $|z_p| > 2$ et $c \notin M$

Démonstration :

Lemme : On considère la suite $z_{n+1} = z_n^2 + c$ dans \mathbb{C} avec $z_0 = 0$ et $c \in \mathbb{C}$. S'il existe $p \in \mathbb{N}$ tel que $|z_p| > 2$ et $|z_p| \geq |c|$ alors $\lim_{n \rightarrow \infty} |z_n| = \infty$.

On a $z_{p+1} = z_p^2 + c$ et grâce à l'inégalité triangulaire

$$|z_{p+1}| \geq ||z_p|^2 - |c||,$$

et puisque $|z_p| > 2$, on a $|z_p|^2 > |z_p|$, et comme $|z_p| \geq |c|$, on a

$$|z_{p+1}| \geq |z_p|^2 - |z_p|,$$

$$|z_{p+1}| \geq r|z_p|,$$

avec $r = |z_p| - 1 > 1$. On montre maintenant par récurrence que

$$H(k) = \{|z_{p+k}| \geq r^k |z_p|\}$$

est vraie pour tout $k > 0$. Supposons $H(k)$ vraie pour un $k > 0$. D'une part on a

$$|z_{p+k}| > |z_p|, |z_{p+k}| > 2, |z_{p+k}| > |c|, |z_{p+k}|^2 > |z_{p+k}|$$

grâce à $r_k > 1$ et aux diverses hypothèses, d'où

$$|z_{p+k+1}| = |z_{p+k}^2 + c|$$

$$|z_{p+k+1}| \geq |z_{p+k}|^2 - |c|$$

$$|z_{p+k+1}| \geq |z_{p+k}|^2 - |z_{p+k}|$$

$$|z_{p+k+1}| \geq |z_{p+k}|(|z_{p+k}| - 1)$$

$$|z_{p+k+1}| \geq r|z_{p+k}|$$

$$|z_{p+k+1}| \geq r^{k+1}|z_p|$$

donc $H(k+1)$ vraie. D'où le résultat annoncé, puisque $\lim_{k \rightarrow \infty} r^k = \infty$

Pour en déduire une propriété pratique permettant de déterminer si un c donné appartient ou pas à M l'ensemble de Mandelbrot, il suffit de voir que si $|c| > 2$, les hypothèses du Lemme sont vérifiées dès $k = 1$ car $z_1 = c$, donc $c \notin M$. Et si $|c| \leq 2$, alors l'existence d'un p tel que $|z_p| > 2$ montre que $c \notin M$ car $|z_p| > |c|$.


3.1.2. Visualisation

On a vu plus haut que dès que le module d'un z_n est strictement plus grand que 2, la suite diverge vers l'infini, et donc c est en dehors de l'ensemble de Mandelbrot. Cela nous permet d'arrêter le calcul pour les points ayant un module strictement supérieur à 2 et qui sont donc en dehors de l'ensemble de Mandelbrot. Pour les points de l'ensemble de Mandelbrot, le calcul n'arrivera jamais à terme, donc il doit être arrêté après un certain nombre d'itérations déterminé par le programme. Il en résulte que l'image affichée n'est qu'une approximation du vrai ensemble. Voici le code source pour représenter l'ensemble de Mandelbrot :


```
x = linspace(-2,1,400);
y = linspace(-1.5,1.5,400);
[X,Y] = meshgrid(x,y);
C = complex(X,Y);
Z = C
G = ones(Z);
for i = 1 : 200
    Z = Z.^2 + C;
    G(abs(Z) < 2) = G(abs(Z) < 2) + 1;
end

c = jetcolormap(201)
set(gcf(),"color_map",c)
plot(x(1),y(1),x($),y($))
Matplot1(G,[x(1) y($) x($) y(1)])
```

Code source de l'ensemble de Mandelbrot



L'ensemble de Mandelbrot



Une partie de l'ensemble Mandelbrot se trouvant dans $[-1.7591, -1.7586] \times [0.0188, 0.0192]$


3.2. Ensemble de Julia

Étant donnés deux nombres complexes, c et z_0 , définissons la suite (z_n) par la relation de récurrence:


$$z_{n+1} = z_n^2 + c.$$

Pour une valeur donnée de c , l'ensemble de Julia correspondant est la frontière de l'ensemble des valeurs initiales z_0 pour lesquelles la suite est bornée. La définition des ensembles de Julia est


relativement proche de celle de l'ensemble de Mandelbrot qui est l'ensemble de toutes les valeurs de c pour lesquelles la suite (z_n) est bornée, en prenant $z_0 = 0$.



L'ensemble de Julia avec $c = 0.3 + 0.5i$



L'ensemble de Julia avec $c = -0.8 + 0.156i$



L'ensemble de Julia avec $c = -0.4 + 0.6i$

3.3. Lien entre l'ensemble de Mandelbrot et de Julia

L'ensemble de Mandelbrot \mathbf{M} peut être défini comme l'ensemble des complexes c pour lesquels l'ensemble de Julia correspondant, $J(c)$, est connexe. Mais, plus précisément, on a (à la limite) identité entre $J(c)$ et \mathbf{M} au voisinage de c , lorsque c appartient à la frontière de \mathbf{M} . Ainsi, il y a une correspondance proche entre la géométrie de l'ensemble de Mandelbrot à un point donné et la structure de l'ensemble de Julia lui correspondant. Par exemple, un point est dans l'ensemble de Mandelbrot exactement là où l'ensemble de Julia correspondant est connexe.